

BMI Measurement Error Regression: Make it shorter

```
/****** bmi3.sas *****/
options linesize=79 pagesize = 500 noovp formdlim='- ' nodate;
title 'BMI and Health: Like bmi2.sas, but try to make it shorter';

data health;
  infile 'bmihealth.data';
  input age1 bmi1 fat1 cholest1 diastol1
        age2 bmi2 fat2 cholest2 diastol2;
        /* fat1 and fat2 are percent body fat */
  age = (age1+age2)/2; bmi = (bmi1+bmi2)/2; fat = (fat1+fat2)/2;
  cholest = (cholest1+cholest2)/2 ; diastol = (diastol1+diastol2)/2;

proc calis cov vardef=n;
  /* Analyze the covariance matrix (Default is corr). Divide by n to
  get true MLE of covariance matrix. */
  title2 'Full Model';
  var age1 -- diastol2; /* Name the observed variables */
  /* Now give simultaneous equations, separated by commas. Latent
  variables begin with F for factor. Error terms begin with
  E for error or D for disturbance. SAS is not case sensitive.
  You must name all the parameters. Optional starting values in
  parentheses may be given after the parameters. */
  lineqs
    Fcholest = gamma11 Fage + gamma12 Fbmi + gamma13 Ffat + e1,
    Fdiastol = gamma21 Fage + gamma22 Fbmi + gamma23 Ffat + e2,
    age1      = Fage + delta11,
    bmi1      = Fbmi + delta12,
    fat1      = Ffat + delta13,
    age2      = Fage + delta21,
    bmi2      = Fbmi + delta22,
    fat2      = Ffat + delta23,
    cholest1  = Fcholest + eps11,
    diastol1  = Fdiastol + eps12,
    cholest2  = Fcholest + eps21,
    diastol2  = Fdiastol + eps22;
  std /* Variances (not standard deviations) will be
  called V-something. Colon means fill in the numbers. */
    Fage Fbmi Ffat e1 e2
    delta11 delta12 delta13 delta21 delta22 delta23
    eps11 eps12 eps21 eps22 = v: ;
  cov /* Covariances: If not mentioned, it's zero. */
    Fage Ffat Fbmi = phi: , e1 e2 = psi: ,
    delta11 delta12 delta13 eps11 eps12 = TED1_: ,
    delta21 delta22 delta23 eps21 eps22 = TED2_: ;
  bounds 0.0 < v1-v15; /* Variances are positive */

/* Note: These std and cov statements will generate warnings saying
"Shorter parameter list than variable list ..." These can be
ignored. There would be no warnings if we specified the
number of parameters, as in 15 * v: and
delta11 delta12 delta13 eps11 eps12 = 10 * TED1_: etc.
But it's more convenient not to have to count parameters. */
```

```

/* Now fit a reduced model to test H0: gamma12 = gamma22 = 0,
meaning BMI is unrelated to either cholesterol or blood pressure
if we allow for age and percent body fat. Use psummary to see
just the minimal output needed for a likelihood ratio test. */

proc calis cov vardef=n psummary;
  title2 'Reduced Model for testing BMI on both DVs at once';
  var age1 -- diastol2;
  lineqs
    Fcholest = gamma11 Fage + gamma12 Fbmi + gamma13 Ffat + e1,
    Fdiastol = gamma21 Fage + gamma22 Fbmi + gamma23 Ffat + e2,
    age1      = Fage + delta11,
    bmi1      = Fbmi + delta12,
    fat1      = Ffat + delta13,
    age2      = Fage + delta21,
    bmi2      = Fbmi + delta22,
    fat2      = Ffat + delta23,
    cholest1  = Fcholest + eps11,
    diastol1  = Fdiastol + eps12,
    cholest2  = Fcholest + eps21,
    diastol2  = Fdiastol + eps22;
  std
    Fage Fbmi Ffat e1 e2
    delta11 delta12 delta13 delta21 delta22 delta23
    eps11 eps12 eps21 eps22 = v: ;
  cov
    Fage Ffat Fbmi = phi: , e1 e2 = psi: ,
    delta11 delta12 delta13 eps11 eps12 = TED1_: ,
    delta21 delta22 delta23 eps21 eps22 = TED2_: ;
  bounds 0.0 < v1-v15;
  lincon gamma12=0, gamma22=0; /* Constrain model to obey this H0 */
  /* Much safer than fitting a model with the variables just missing
  in the regression equations. However, SAS will warn us that
  "There are 2 active constraints at the solution," and carry on
  in a pretty menacing way. The warning can be ignored, in this
  particular case where the linear constraints are setting
  parameters exactly equal to zero. Do NOT ignore the warning
  if an inequality is involved. */

proc iml;
  title2 'Calculate Likelihood ratio test of H0: gamma12=gamma22=0';
  G = (19.5309 - 18.9011) * 500/499 ;
  /* Difference between chisquares (corrected) */
  pval = 1 - probchi(G,2);
  print G,pval;

```

BMI and Health: Like bmi2.sas, but try to make it shorter 1
Full Model

The CALIS Procedure
Covariance Structure Analysis: Pattern and Initial Values

LINEQS Model Statement

		Matrix	Rows	Columns	-----Matrix Type-----	
Term 1	1	_SEL_	10	27	SELECTION	
	2	_BETA_	27	27	EQSBETA	IMINUSINV
	3	_GAMMA_	27	15	EQSGAMMA	
	4	_PHI_	15	15	SYMMETRIC	

The 12 Endogenous Variables

Manifest	age1	bmi1	fat1	cholest1	diastol1	age2
	bmi2	fat2	cholest2	diastol2		
Latent	Fcholest	Fdiastol				

The 15 Exogenous Variables

Manifest						
Latent	Fage	Fbmi	Ffat			
Error	e1	e2	eps11	eps12	eps21	
	eps22	delta11	delta12	delta13	delta21	
	delta22	delta23				

BMI and Health: Like bmi2.sas, but try to make it shorter 2
Full Model

The CALIS Procedure
Covariance Structure Analysis: Pattern and Initial Values

Manifest Variable Equations with Initial Estimates

age1	=	1.0000	Fage	+	1.0000	delta11
bmi1	=	1.0000	Fbmi	+	1.0000	delta12
fat1	=	1.0000	Ffat	+	1.0000	delta13
cholest1	=	1.0000	Fcholest	+	1.0000	eps11
diastol1	=	1.0000	Fdiastol	+	1.0000	eps12
age2	=	1.0000	Fage	+	1.0000	delta21
bmi2	=	1.0000	Fbmi	+	1.0000	delta22
fat2	=	1.0000	Ffat	+	1.0000	delta23
cholest2	=	1.0000	Fcholest	+	1.0000	eps21
diastol2	=	1.0000	Fdiastol	+	1.0000	eps22

The CALIS Procedure
Covariance Structure Analysis: Pattern and Initial Values

Latent Variable Equations with Initial Estimates

$$\begin{aligned} \text{Fcholest} = & \quad . * \text{Fage} \quad + \quad . * \text{Fbmi} \quad + \quad . * \text{Ffat} \\ & \quad \text{gamma11} \quad \quad \quad \text{gamma12} \quad \quad \quad \text{gamma13} \\ & \quad \quad \quad + \quad 1.0000 \text{ e1} \end{aligned}$$

$$\begin{aligned} \text{Fdiastol} = & \quad . * \text{Fage} \quad + \quad . * \text{Fbmi} \quad + \quad . * \text{Ffat} \\ & \quad \text{gamma21} \quad \quad \quad \text{gamma22} \quad \quad \quad \text{gamma23} \\ & \quad \quad \quad + \quad 1.0000 \text{ e2} \end{aligned}$$

Variances of Exogenous Variables

Variable	Parameter	Estimate
Fage	v1	.
Fbmi	v2	.
Ffat	v3	.
e1	v4	.
e2	v5	.
eps11	v12	.
eps12	v13	.
eps21	v14	.
eps22	v15	.
delta11	v6	.
delta12	v7	.
delta13	v8	.
delta21	v9	.
delta22	v10	.
delta23	v11	.

Covariances Among Exogenous Variables

Var1	Var2	Parameter	Estimate
Fage	Fbmi	phi2	.
Fage	Ffat	phi1	.
Fbmi	Ffat	phi3	.
e1	e2	psi1	.
eps11	eps12	TED1_10	.
eps21	eps22	TED2_10	.
eps11	delta11	TED1_4	.
eps12	delta11	TED1_7	.
eps11	delta12	TED1_5	.
eps12	delta12	TED1_8	.

```

delta11  delta12  TED1_1      .
eps11    delta13  TED1_6      .
eps12    delta13  TED1_9      .
delta11  delta13  TED1_2      .
delta12  delta13  TED1_3      .
eps21    delta21  TED2_4      .
eps22    delta21  TED2_7      .
eps21    delta22  TED2_5      .
eps22    delta22  TED2_8      .
delta21  delta22  TED2_1      .
eps21    delta23  TED2_6      .
eps22    delta23  TED2_9      .
delta21  delta23  TED2_2      .
delta22  delta23  TED2_3      .

```

BMI and Health: Like bmi2.sas, but try to make it shorter 4
Full Model

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Observations	500	Model Terms	1
Variables	10	Model Matrices	4
Informations	55	Parameters	45

Variable	Mean	Std Dev
age1	43.08800	12.99955
bmi1	25.43580	4.54927
fat1	18.79000	7.75224
cholest1	262.02740	52.50262
diastol1	88.42400	19.27206
age2	44.40600	12.41762
bmi2	25.51760	3.72546
fat2	18.90280	7.56396
cholest2	261.24560	53.87560
diastol2	88.70000	13.08610

NOTE: Some initial estimates computed by two-stage LS method.

Now we get the (automatic) starting values of all the parameters. Skip it.

Optimization Start

Active Constraints 0 Objective Function 0.3271903962
 Max Abs Gradient Element 0.0798909029

Iter	Rest arts	Func Calls	Act Con	Objective Function	Obj Fun Change	Max Abs Gradient Element	Step Size	Slope Search Direc
1	0	3	0	0.30488	0.0223	0.2450	0.0467	-1.031
2	0	4	0	0.28493	0.0200	0.1155	0.321	-0.174
3	0	5	0	0.26868	0.0163	0.0924	1.000	-0.0414
4	0	7	0	0.21790	0.0508	0.2954	2.854	-0.0371
5	0	8	0	0.19763	0.0203	0.1178	1.000	-0.0295
6	0	10	0	0.18967	0.00796	0.0603	1.110	-0.0139
7	0	12	0	0.18633	0.00334	0.0332	1.482	-0.0039
8	0	13	0	0.18070	0.00563	0.0151	1.517	-0.0055

It goes on and on for 164 iterations. Skipping ...

Optimization Results

Iterations 164 Function Calls 293
 Gradient Calls 232 Active Constraints 0
 Objective Function 0.0378779129 Max Abs Gradient Element 9.0354545E-6
 Slope of Search Direction -5.202293E-7

ABSGCONV convergence criterion satisfied.

 BMI and Health: Like bmi2.sas, but try to make it shorter 7
 Full Model

The CALIS Procedure
 Covariance Structure Analysis: Maximum Likelihood Estimation

Fit Function	0.0379
Goodness of Fit Index (GFI)	0.9926
GFI Adjusted for Degrees of Freedom (AGFI)	0.9591
Root Mean Square Residual (RMR)	19.4718
Parsimonious GFI (Mulaik, 1989)	0.2206
Chi-Square	18.9011
Chi-Square DF	10
Pr > Chi-Square	0.0415
Independence Model Chi-Square	4015.9
Independence Model Chi-Square DF	45
RMSEA Estimate	0.0422

Skipping again ...

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Latent Variable Equations with Estimates

Fcholest =	0.1480*Fage	+	-1.2703*Fbmi	+	2.2271*Ffat
Std Err	0.2128 gamma11		1.6550 gamma12		0.8974 gamma13
t Value	0.6957		-0.7675		2.4818

+ 1.0000 e1

Fdiastol =	0.0104*Fage	+	-0.1054*Fbmi	+	1.2692*Ffat
Std Err	0.0458 gamma21		0.3941 gamma22		0.2044 gamma23
t Value	0.2268		-0.2675		6.2090

+ 1.0000 e2

Variances of Exogenous Variables

Variable	Parameter	Estimate	Standard Error	t Value
Fage	v1	148.12838	9.79411	15.12
Fbmi	v2	12.54181	0.94050	13.34
Ffat	v3	46.99750	3.28849	14.29
e1	v4	2391	161.53168	14.81
e2	v5	75.69775	8.78257	8.62
eps11	v12	195.80618	54.03321	3.62
eps12	v13	207.71308	15.24803	13.62
eps21	v14	336.03140	56.74542	5.92
eps22	v15	25.20654	7.67288	3.29
delta11	v6	22.30517	3.22022	6.93
delta12	v7	8.61051	0.71704	12.01
delta13	v8	13.95514	1.71930	8.12
delta21	v9	6.30433	2.90105	2.17
delta22	v10	1.28485	0.48547	2.65
delta23	v11	11.46363	1.67508	6.84

Covariances Among Exogenous Variables

Var1	Var2	Parameter	Estimate	Standard Error	t Value
Fage	Fbmi	phi2	8.53524	2.14246	3.98
Fage	Ffat	phi1	28.67293	4.20356	6.82
Fbmi	Ffat	phi3	21.07021	1.58695	13.28
e1	e2	psi1	26.99445	23.13516	1.17
eps11	eps12	TED1_10	2.37369	17.44389	0.14
eps21	eps22	TED2_10	10.55839	12.14484	0.87

eps11	delta11	TED1_4	1.46571	9.28626	0.16
eps12	delta11	TED1_7	5.50639	4.26027	1.29
eps11	delta12	TED1_5	-1.68780	4.05537	-0.42
eps12	delta12	TED1_8	8.22814	2.38542	3.45
delta11	delta12	TED1_1	2.41139	1.00109	2.41
eps11	delta13	TED1_6	-1.67949	6.62998	-0.25
eps12	delta13	TED1_9	0.90570	3.57876	0.25
delta11	delta13	TED1_2	-1.23726	1.62183	-0.76
delta12	delta13	TED1_3	8.07727	0.96844	8.34
eps21	delta21	TED2_4	0.83261	9.02367	0.09
eps22	delta21	TED2_7	2.24903	2.78751	0.81
eps21	delta22	TED2_5	0.09756	3.34404	0.03
eps22	delta22	TED2_8	2.97256	1.44467	2.06
delta21	delta22	TED2_1	0.26655	0.76726	0.35
eps21	delta23	TED2_6	-4.03672	6.65942	-0.61
eps22	delta23	TED2_9	2.33466	2.56252	0.91
delta21	delta23	TED2_2	1.97235	1.51921	1.30
delta22	delta23	TED2_3	0.06895	0.74323	0.09

Skip the “standardized estimates.” Next comes the output from the reduced model. The `psummary` option makes it very short. Note the scary warnings, which can be ignored.

BMI and Health: Like `bmi2.sas`, but try to make it shorter 12
 Reduced Model for testing BMI on both DVs at once

The CALIS Procedure
 Covariance Structure Analysis: Maximum Likelihood Estimation

WARNING: There are 2 active constraints at the solution. The standard errors and Chi-Square test statistic assume the solution is located in the interior of the parameter space and hence do not apply if it is likely that some different set of inequality constraints could be active.

NOTE: The degrees of freedom are increased by the number of active constraints (see Dijkstra, 1992). The number of parameters in calculating fit indices is decreased by the number of active constraints. To turn off the adjustment, use the `NOADJDF` option.

BMI and Health: Like bmi2.sas, but try to make it shorter
Reduced Model for testing BMI on both DVs at once

13

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Fit Function	0.0391
Goodness of Fit Index (GFI)	0.9924
GFI Adjusted for Degrees of Freedom (AGFI)	0.9650
Root Mean Square Residual (RMR)	19.7718
Parsimonious GFI (Mulaik, 1989)	0.2646
Chi-Square	19.5309
Chi-Square DF	12
Pr > Chi-Square	0.0765
Independence Model Chi-Square	4015.9
Independence Model Chi-Square DF	45
RMSEA Estimate	0.0355
RMSEA 90% Lower Confidence Limit	.
RMSEA 90% Upper Confidence Limit	0.0630
ECVI Estimate	0.2154
ECVI 90% Lower Confidence Limit	.
ECVI 90% Upper Confidence Limit	0.2482
Probability of Close Fit	0.7834
Bentler's Comparative Fit Index	0.9981
Normal Theory Reweighted LS Chi-Square	19.2565
Akaike's Information Criterion	-4.4691
Bozdogan's (1987) CAIC	-67.0444
Schwarz's Bayesian Criterion	-55.0444
McDonald's (1989) Centrality	0.9925
Bentler & Bonett's (1980) Non-normed Index	0.9929
Bentler & Bonett's (1980) NFI	0.9951
James, Mulaik, & Brett (1982) Parsimonious NFI	0.2654
Z-Test of Wilson & Hilferty (1931)	1.4315
Bollen (1986) Normed Index Rho1	0.9818
Bollen (1988) Non-normed Index Delta2	0.9981
Hoelter's (1983) Critical N	539

BMI and Health: Like bmi2.sas, but try to make it shorter
Calculate Likelihood ratio test of H0: gamma12=gamma22=0

14

G

0.6310621

PVAL

0.7294014